Práctica 1

Determinantes. Sistemas de ecuaciones. Dependencia lineal

1.1. Enunciados de Prácticas

Práctica A Calcular el determinante de las siguientes matrices:

$$(B)\begin{pmatrix} 6 & -5 & 8 & 4 \\ 9 & 7 & 5 & 2 \\ 7 & 5 & 3 & 7 \\ -4 & 8 & -8 & -3 \end{pmatrix};\quad (I)\begin{pmatrix} 2 & -1 & 0 & 0 & 0 & 0 \\ -1 & 2 & -1 & 0 & 0 & 0 \\ 0 & -1 & 2 & -1 & 0 & 0 \\ 0 & 0 & -1 & 2 & -1 & 0 \\ 0 & 0 & 0 & -1 & 2 & -1 \\ 0 & 0 & 0 & 0 & -1 & 2 \end{pmatrix}$$

Práctica B Calcular la inversa de las siguientes matrices:

$$\begin{pmatrix} 2 & 7 & 3 \\ 3 & 9 & 4 \\ 1 & 5 & 3 \end{pmatrix},\quad \begin{pmatrix} 1 & 2 & 3 & 4 \\ 2 & 3 & 1 & 2 \\ 1 & 1 & 1 & -1 \\ 1 & 0 & -2 & -6 \end{pmatrix}$$

Práctica C Resolver los sistemas:

$$(C)\left\{\begin{array}{rcl} 2x_1+2x_2-x_3+x_4 & = & 4 \\ 4x_1+3x_2-x_3+2x_4 & = & 6 \\ 8x_1+5x_2-3x_3+4x_4 & = & 12 \\ 3x_1+3x_2-2x_3+2x_4 & = & 6 \end{array}\right. ;$$

$$(E)\left\{\begin{array}{rcl} 0.63x_1+1.00x_2+0.71x_3+0.34x_4 & = & 2.08 \\ 1.17x_1+0.18x_2-0.65x_3+0.71x_4 & = & 0.17 \\ 2.71x_1-0.75x_2+1.17x_3-2.35x_4 & = & 1.28 \\ 3.58x_1+0.28x_2-3.45x_3-1.18x_4 & = & 0.05 \end{array}\right.$$

Práctica D Indicando los pasos de la eliminación gaussiana, volver a resolver los sistemas de la práctica anterior.

Práctica E Consideramos el sistema de vectores de $\mathbb{R}^4$ dado por $\mathcal{S} = \{f_1, f_2, f_3, f_4, f_5\}$, donde:

$$f_1 = (1, -1, 3, 2);\ f_2 = (0, 1, 2, -1);\ f_3 = (2, -3, 4, 5)$$
$$f_4 = (1, 0, 4, 1);\ f_5 = (1, -3, 0, 4)$$

Se pide:

1. Calcular el rango de $\mathcal{S}$.
2. Dar una base de $H = \langle f_1, f_2, f_3, f_4, f_5 \rangle$.
3. Encontrar, si es que existe, una combinación no trivial $\alpha_1 f_1 + \alpha_2 f_2 + \alpha_3 f_3 + \alpha_4 f_4 + \alpha_5 f_5 = 0$.

Práctica F Consideramos las matrices de $M_{2\times 2}(\mathbb{R})$ dadas por las igualdades:

$$m_1 = \begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix};\quad m_2 = \begin{pmatrix} -1 & 3 \\ 1 & 4 \end{pmatrix};\quad m_3 = \begin{pmatrix} 2 & 1 \\ 7 & 2 \end{pmatrix};$$
$$m_4 = \begin{pmatrix} 6 & 1 \\ 16 & 4 \end{pmatrix};\quad m = \begin{pmatrix} -15 & 11 \\ -36 & 14 \end{pmatrix};$$

y el subespacio $H = \langle m_1, m_2, m_3, m_4 \rangle$. Se pide:

1. Calcular $\dim H$ y una base de H.
2. Decidir si existe una combinación no trivial $\alpha_1 m_1 + \alpha_2 m_2 + \alpha_3 m_3 + \alpha_4 m_4 = 0$ y en caso afirmativo calcular una de ellas.
3. Comprobar que $m \in H$ y calcular las coordenadas de m respecto a la base dada en el punto 1 anterior.

Práctica G Consideramos el sistema cuadrado de orden n definido por la igualdad:

$$\begin{bmatrix} 2 & -1 & & & \\ -1 & 2 & -1 & & \\ & -1 & 2 & -1 & \\ & & \ddots & \ddots & \ddots \\ & & & -1 & 2 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ \vdots \\ x_n \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ \vdots \\ 1 \end{bmatrix}$$

Se pide:

1. Resolverlo cuando $n = 4, 6, 10$.
2. A partir del anterior punto, deducir una fórmula para la solución del sistema anterior válida para cualquier n.

Práctica H ([Fr], sección de estática del sólido rígido) De una barra de masa despreciable que está sujeta por su extremo O sabemos que soporta un peso P a una distancia x de O y que permanece en equilibrio en posición horizontal gracias a la acción de una cuerda que sujeta la barra a una distancia y de O formando un ángulo θ con la vertical. Se pide calcular la tensión F soportada por la cuerda.

Para realizar este problema y el siguiente ha de tenerse en cuenta la ley de la palanca y la ley de Hooke: la fuerza ejercida por un resorte elástico está dirigida hacia la posición de equilibrio y es proporcional a la desviación a partir de dicha posición de equilibrio.



Figura 1.1: Equilibrio de una barra.

Práctica I ([Fr], sección de estática del sólido rígido) Consideramos una varilla delgada de masa m y longitud L que pende de dos muelles elásticos verticales de constantes k_1, k_2 y de longitudes l_{01}, l_{02} sin deformar, situados a distancias d_1, d_2 a uno y otro lado del centro de masas, O, de la varilla. Se pide calcular las longitudes l_1, l_2 a las que ha de colgarse la varilla para que está permanezca en equilibrio en posición horizontal.



Figura 1.2: Varilla que pende de dos muelles.

Práctica J (ver [Na], pág. 106) En la figura 1.3.iz se muestra un circuito eléctrico conectado a tres terminales con voltajes conocidos. Obténgase los voltajes en los nodos a, b y c.

Para resolver este problema, téngase en cuenta la primera *Ley de Kirchhoff*: 'La suma de todas las intensidades de corriente que llegan a un nodo (unión o empalme) de un circuito cerrado es igual a la suma de todas las intensidades de corriente que salen de él'. Así, para el nodo de la figura 1.3.dr debe ocurrir la fórmula

$$i_{ab} + i_{ac} + \cdots + i_{ak} = 0$$

para $i_{ab} = \frac{Va-Vb}{r_{ab}}$ la intensidad de corriente que circula por ab, r_{ab} la magnitud de la resistencia ab y V_a, V_b los voltajes en a y b, respectivamente.



Figuras 1.3: (iz) Circuito eléctrico con resistencias.
(dr) Nodo a conectado a resistencias b, c, d,..., k.

1.2. Elementos necesarios

Cargamos el paquete de Álgebra Lineal

```
> with(linalg):
```

```
Warning, new definition for norm
```

1.3. Solución de la práctica A

```
> Ab:=matrix(4,4,[[6,-5,8,4],[9,7,5,2],[7,5,3,7],[-4,8,-8,-3]]);
```

$$Ab := \begin{bmatrix} 6 & -5 & 8 & 4 \\ 9 & 7 & 5 & 2 \\ 7 & 5 & 3 & 7 \\ -4 & 8 & -8 & -3 \end{bmatrix}$$

```
> SolAb:=det(Ab);
```

$$SolAb := 100$$

```
> Ai:=band([-1,2,-1],6);
```

$$\mathrm{Ai} := \begin{bmatrix} 2 & -1 & 0 & 0 & 0 & 0 \\ -1 & 2 & -1 & 0 & 0 & 0 \\ 0 & -1 & 2 & -1 & 0 & 0 \\ 0 & 0 & -1 & 2 & -1 & 0 \\ 0 & 0 & 0 & -1 & 2 & -1 \\ 0 & 0 & 0 & 0 & -1 & 2 \end{bmatrix}$$

```
> SolAi:=det(Ai);
```

$$SolAi := 7$$

1.4. Solución de la práctica B

```
> B1:=matrix(3,3,[[2,7,3],[3,9,4],[1,5,3]]);
```

$$B1 := \begin{bmatrix} 2 & 7 & 3 \\ 3 & 9 & 4 \\ 1 & 5 & 3 \end{bmatrix}$$

```
> SolB1:=inverse(B1);
```

$$SolB1 := \begin{bmatrix} \frac{-7}{3} & 2 & \frac{-1}{3} \\ \frac{5}{3} & -1 & \frac{-1}{3} \\ -2 & 1 & 1 \end{bmatrix}$$

1.5. Solución de la práctica C

```
> Cc:=matrix(4,4,[[2,2,-1,1],[4,3,-1,2],[8,5,-3,4],[3,3,-2,2]]);
```

$$Cc := \begin{bmatrix} 2 & 2 & -1 & 1 \\ 4 & 3 & -1 & 2 \\ 8 & 5 & -3 & 4 \\ 3 & 3 & -2 & 2 \end{bmatrix}$$

```
> CcIn:=vector([4,6,12,6]);
```

$$CcIn := [4,\ 6,\ 12,\ 6]$$

```
> SolCc:=linsolve(Cc,CcIn);
```

$$SolCc := [1,\ 1,\ -1,\ -1]$$

```
> Ce:=matrix(4,4,[[0.63,1,0.71,0.34],[1.17,0.18,-0.65,0.71],[2.71,-0.75
> ,1.17,-2.35],
> [3.58,0.28,-3.45,-1.18]]);
```

$$Ce := \begin{bmatrix} 0.63 & 1 & 0.71 & 0.34 \\ 1.17 & 0.18 & -0.65 & 0.71 \\ 2.71 & -0.75 & 1.17 & -2.35 \\ 3.58 & 0.28 & -3.45 & -1.18 \end{bmatrix}$$

```
> CeIn:=vector([2.08,0.17,1.28,0.05]);
```

$$CeIn := [2.08,\ 0.17,\ 1.28,\ 0.05]$$

```
> SolCe:=linsolve(Ce,CeIn);
```

$$SolCe := [0.4053636920,\ 1.475019770,\ 0.6094906496,\ -0.2445213457]$$

1.6. Solución de la práctica D

```
> Dc:=matrix(4,5,[[2,2,-1,1,4],[4,3,-1,2,6],[8,5,-3,4,12],[3,3,-2,2,6]]
> );
```

$$Dc := \begin{bmatrix} 2 & 2 & -1 & 1 & 4 \\ 4 & 3 & -1 & 2 & 6 \\ 8 & 5 & -3 & 4 & 12 \\ 3 & 3 & -2 & 2 & 6 \end{bmatrix}$$

```
> DcP:=pivot(Dc,1,1); print(`etapa 1`);
```

$$DcP := \begin{bmatrix} 2 & 2 & -1 & 1 & 4 \\ 0 & -1 & 1 & 0 & -2 \\ 0 & -3 & 1 & 0 & -4 \\ 0 & 0 & \frac{-1}{2} & \frac{1}{2} & 0 \end{bmatrix}$$

etapa 1

```
> DcP:=pivot(DcP,2,2); print(`etapa 2`);
```

$$DcP := \begin{bmatrix} 2 & 0 & 1 & 1 & 0 \\ 0 & -1 & 1 & 0 & -2 \\ 0 & 0 & -2 & 0 & 2 \\ 0 & 0 & \frac{-1}{2} & \frac{1}{2} & 0 \end{bmatrix}$$

etapa 2

```
> DcP:=pivot(DcP,3,3); print(`etapa 3`);
```

$$DcP := \begin{bmatrix} 2 & 0 & 0 & 1 & 1 \\ 0 & -1 & 0 & 0 & -1 \\ 0 & 0 & -2 & 0 & 2 \\ 0 & 0 & 0 & \frac{1}{2} & \frac{-1}{2} \end{bmatrix}$$

etapa 3

```
> DcP:=pivot(DcP,4,4); print(`etapa 4`);
```

$$DcP := \begin{bmatrix} 1 & 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & -1 \end{bmatrix}$$

etapa 4

```
> SolDc:=backsub(DcP);
```

$$SolDc := [1, 1, -1, -1]$$

Se pueden hacer todas las etapas a la vez utilizando el comando **hermite** que además tiene las siguientes ventajas: 1) nos da la matriz F de la transformación en filas y 2) es aplicable incluso cuando los coeficientes de la matriz son polinomios en una variable x, por eso es necesario decirle al programa el nombre de la variable independiente.

Utilizando el comando **gausselim**, gausselim(Dc), también conseguimos hacer todas las etapas a la vez. Este comando no tiene las dos ventajas que expresábamos anteriormente.

```
> DcP:=hermite(Dc,x,'U'); U:=evalm(U);
```

$$DcP := \begin{bmatrix} 1 & 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & -1 \end{bmatrix}$$

$$U := \begin{bmatrix} 1 & \frac{-1}{2} & \frac{1}{2} & -1 \\ 1 & \frac{1}{2} & \frac{-1}{2} & 0 \\ -1 & \frac{3}{2} & \frac{-1}{2} & 0 \\ -4 & \frac{3}{2} & \frac{-1}{2} & 2 \end{bmatrix}$$

Comprobación de que $U \cdot Dc = DcP$

```
> evalm(U&*Dc=DcP);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & -1 \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & -1 \end{bmatrix}$$

1.7. Solución de la práctica E

```
> E:=matrix(5,4,[[1,-1,3,2],[0,1,2,-1],[2,-3,4,5],[1,0,4,1],[1,-3,0,4]]
> );
```

$$E := \begin{bmatrix} 1 & -1 & 3 & 2 \\ 0 & 1 & 2 & -1 \\ 2 & -3 & 4 & 5 \\ 1 & 0 & 4 & 1 \\ 1 & -3 & 0 & 4 \end{bmatrix}$$

```
> SolE1:=rank(E);
```

$$SolE1 := 3$$

```
> E2:=hermite(E,x,'U'); U:=evalm(U);
```

$$E2 := \begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

$$U := \begin{bmatrix} -4 & -4 & 0 & 5 & 0 \\ -2 & -1 & 0 & 2 & 0 \\ 1 & 1 & 0 & -1 & 0 \\ -2 & 1 & 1 & 0 & 0 \\ -2 & 1 & 0 & 1 & 1 \end{bmatrix}$$

```
> evalm(U&*E=E2);
```

$$\begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

SolE2={ F1=(1,0,0,1); F2=(0,1,0,-1); F3=(0,0,1,0)}

SolE3={ α1= -2, α2= 1, α3= 1, α4= 0, α5= 0}.
Ponemos como solución de $E3$ la cuarta fila de la matriz U, también podríamos dar como solución la quinta fila de U, es decir, $\{\alpha_1 = -2, \alpha_2 = 1, \alpha_3 = 0, \alpha_4 = 1, \alpha_5 = 1\}$.

1.8. Solución de la práctica F

```
> F1:=matrix(4,4,[[1,2,3,4],[-1,3,1,4],[2,1,7,2],[6,1,16,4]]);
```

$$F1 := \begin{bmatrix} 1 & 2 & 3 & 4 \\ -1 & 3 & 1 & 4 \\ 2 & 1 & 7 & 2 \\ 6 & 1 & 16 & 4 \end{bmatrix}$$

```
> SolF1:=rank(F1);
```

$$SolF1 := 3$$

```
> F2:=hermite(F1,x,'U'); U:=eval(U);
```

$$F2 := \begin{bmatrix} 1 & 0 & 0 & \frac{22}{17} \\ 0 & 1 & 0 & \frac{32}{17} \\ 0 & 0 & 1 & \frac{-6}{17} \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

$$U := \begin{bmatrix} \frac{20}{17} & \frac{-11}{17} & \frac{-7}{17} & 0 \\ \frac{9}{17} & \frac{1}{17} & \frac{-4}{17} & 0 \\ \frac{-7}{17} & \frac{3}{17} & \frac{5}{17} & 0 \\ -1 & 1 & -2 & 1 \end{bmatrix}$$

Comprobación de que m pertenece a H

```
> eval(-15*22/17+11*32/17-36*(-6/17)=14);
```

$$14 = 14$$

Las tres primeras filas de F_2 constituyen una base de H.

Relación entre m1,m2,m3,m4 la dada por α1= -1, α2= 1, α3= -2, α4= 1

Coordenadas de m=(-15, 11, -36)

1.9. Solución de la práctica G

```
> nn:=4;
```

$$nn := 4$$

```
> G:=band([-1,2,-1],nn); Gb:=seq(0,i=1..nn-1): Gb:=[Gb,1];
```

$$G := \begin{bmatrix} 2 & -1 & 0 & 0 \\ -1 & 2 & -1 & 0 \\ 0 & -1 & 2 & -1 \\ 0 & 0 & -1 & 2 \end{bmatrix}$$

$$Gb := [0, 0, 0, 1]$$

```
> SolG:=linsolve(G,Gb);
```

$$SolG := \left[\frac{1}{5}, \frac{2}{5}, \frac{3}{5}, \frac{4}{5}\right]$$

Se repite la ejecución de las tres ordenes anteriores cambiando nn por 6 y 10 y se obtiene toda la solución al apartado 1

1.10. Solución de la práctica H

Por la ley de la palanca debe ser $P \cdot x = F \cos\theta \cdot y$, luego

$$F = \frac{xP}{y\cos\theta}$$

1.11. Solución de la práctica I

Para que la varilla esté en equilibrio en posición horizontal deberán ocurrir las siguientes condiciones:

- El balance de fuerzas sobre la varilla es cero. Si denotamos por x_1 la magnitud de la desviación de la posición de equilibrio del primero de los muelles por efecto del peso de la varilla; entonces, $l_1 = l_{01}+x_1$ y $|F_1| = k_1x_1$. La condición expresada, en consecuencia, se escribe:

$$C_1\colon\, k_1x_1 + k_2x_2 = mg$$

- El momento resultante respecto al centro de masas es cero. O lo que es lo mismo, por la ley de la palanca, se verifica:

$$C_2\colon\, k_1x_1d_1 = k_2x_2d_2$$

Resolviendo las ecuaciones C_1, C_2 obtenemos

$$x_1 = \frac{mg}{k_1}\frac{d_2}{d_1+d_2}; \qquad x_2 = \frac{mg}{k_2}\frac{d_1}{d_1+d_2}$$

y, por tanto, que las longitudes a las que ha de colgarse la varilla verifican:

$$l_1 = l_{01} + \frac{mg}{k_1}\frac{d_2}{d_1+d_2}; \qquad l_2 = l_{02} + \frac{mg}{k_2}\frac{d_1}{d_1+d_2}$$

Todo lo anterior lo podemos resolver con el listado:

```
> C1:=k1*x1+k2*x2=m*g;
> C2:=k1*x1*d1=k2*x2*d2;
```

$$C1 := k1\ x1 + k2\ x2 = m\, g$$

$$C2 := k1\ x1\ d1 = k2\ x2\ d2$$

```
> solve({C1,C2},{x1,x2});
```

$$\{x1 = \frac{m\, g\, d2}{(d1 + d2)\, k1},\ x2 = \frac{d1\, m\, g}{k2\, (d1 + d2)}\}$$

1.12. Solución de la práctica J

```
> EcA:=(Va-20)/2+(Va-Vb)/4+(Va-Vc)/3=0;
> EcB:=(Vb-Va)/4+(Vb-0)/3+(Vb-Vc)/5=0;
> EcC:=(Vc-5)/3+(Vc-Va)/3+(Vc-Vb)/5=0;
> Ecs:={EcA,EcB,EcC};
```

$$EcA := \frac{13}{12}\ Va - 10 - \frac{1}{4}\ Vb - \frac{1}{3}\ Vc = 0$$

$$EcB := \frac{47}{60}\ Vb - \frac{1}{4}\ Va - \frac{1}{5}\ Vc = 0$$

$$EcC := \frac{13}{15}\ Vc - \frac{5}{3} - \frac{1}{3}\ Va - \frac{1}{5}\ Vb = 0$$

$$Ecs := \{\frac{13}{15}\ Vc - \frac{5}{3} - \frac{1}{3}\ Va - \frac{1}{5}\ Vb = 0,\ \frac{13}{12}\ Va - 10 - \frac{1}{4}\ Vb - \frac{1}{3}\ Vc = 0,$$
$$\frac{47}{60}\ Vb - \frac{1}{4}\ Va - \frac{1}{5}\ Vc = 0\}$$

```
> SolH:=solve(Ecs,{Va,Vb,Vc}); SolH:=evalf(SolH,4);
```

$$SolH := \{Vc = \frac{4775}{559},\ Vb = \frac{3600}{559},\ Va = \frac{7460}{559}\}$$
$$SolH := \{Vb = 6.440,\ Va = 13.35,\ Vc = 8.542\}$$

Las unidades de Va, Vb, Vc, en voltios.

```
> HM:=genmatrix(Ecs,[Va,Vb,Vc],flag);
```

$$HM := \begin{bmatrix} \frac{-1}{3} & \frac{-1}{5} & \frac{13}{15} & \frac{5}{3} \\ \frac{13}{12} & \frac{-1}{4} & \frac{-1}{3} & 10 \\ \frac{-1}{4} & \frac{47}{60} & \frac{-1}{5} & 0 \end{bmatrix}$$

HM es la matriz ampliada del sistema de ecuaciones *Ecs*.

1.13. Ejercicios

Práctica K Como datos de la práctica I tenemos:

- Constantes elásticas de los muelles: $k_1 = 30\,[\![\mathrm{N/m}]\!]$, $k_2 = 15\,[\![\mathrm{N/m}]\!]$.
- Masa de la barra, $m = 1\,[\![\mathrm{kg}]\!]$.
- Longitudes de los muelles sin deformar, $l_{01} = l_{02} = 0.3\,[\![\mathrm{m}]\!]$.
- Distancias al centro de masas, $d_1 = 0.3\,[\![\mathrm{m}]\!]$, $d_2 = 0.6\,[\![\mathrm{m}]\!]$.

Calcular las longitudes l_1, l_2 a las que ha de colgarse la varilla para que permanezca en equilibrio en posición horizontal.

Sol. $l_1 = l_2 = 0.518\,[\![\mathrm{m}]\!]$

Práctica L Como datos de la práctica I tenemos:

- Masa de la barra, $m = 1\,[\![\mathrm{kg}]\!]$.
- Longitudes de los muelles sin deformar, $l_{01} = l_{02} = 0.3\,[\![\mathrm{m}]\!]$.
- Las longitudes totales a las que permanece la varilla en equilibrio y en posición horizontal, $l_1 = 0.45\,[\![\mathrm{m}]\!], l_2 = 0.55\,[\![\mathrm{m}]\!]$.
- Distancias al centro de masas, $d_1 = d_2 = 0.5\,[\![\mathrm{m}]\!]$.

Calcular las constantes de Hooke k_1, k_2 de cada uno de los muelles.

Sol. $k_1 = 32.667\,[\![\mathrm{N/m}]\!]$, $k_2 = 19.600\,[\![\mathrm{N/m}]\!]$

Práctica M Como datos de la práctica H tenemos:

- Nuestra barra soporta tres pesas de 30 $[\![\mathrm{g}]\!]$, 50 $[\![\mathrm{g}]\!]$ y 20 $[\![\mathrm{g}]\!]$ situadas a 30 $[\![\mathrm{cm}]\!]$, 40 $[\![\mathrm{cm}]\!]$ y 60 $[\![\mathrm{cm}]\!]$ de O, respectivamente.
- La cuerda sujeta a la barra a una distancia de 20 $[\![\mathrm{cm}]\!]$ de O y forma con la vertical un ángulo de 30^o.

Calcular la tensión que soporta la cuerda.

Sol. $F = 2.320\,[\![\mathrm{N}]\!]$